HYPERPLANE ARRANGEMENTS

Questions

(1) A subset $J = \{j_1, \dots, j_p\} \subseteq [n] := \{1, \dots, n\}$ is a *circuit* if $(H_{j_1}, \dots, H_{j_p})$ is minimally dependent. A subset $J \subseteq [n]$ is a *broken circuit* if there exists j_{p+1} with $j_i < j_{p+1}$ for $1 \le i \le p$ and $J \cup \{j_{p+1}\}$ is a circuit.

(a) Suppose that $Q(\mathcal{A}) = (x+y-z)xyz(x+y)$. Determine all the broken circuits for $\mathcal{A}$.

(b) Using `Macaulay2`, find a basis for the p-th graded component of the Orlik-Solomon algebra $\big(A(\mathcal{A})\big)_p$ for $0 \le p \le 3$.

(c) What is the relationship between (a) and (b)? Can you prove it?

(2) Let $\mathcal{A}$ be an arrangement and let $L = L(\mathcal{A})$. Define the *Möbius function* $\mu\colon L \times L \to \mathbb{Z}$ as follows:

$$\begin{aligned} \mu(X,X) &= 1 && \text{if } X \in L, \\ \textstyle\sum_{Z : X \le Z \le Y} \mu(X,Z) &= 0 && \text{if } X < Y, \\ \mu(X,Y) &= 0 && \text{otherwise.} \end{aligned}$$

For $X \in L$, define $\mu(X) = \mu(V,X)$. The *Poincaré polynomial* of $\mathcal{A}$ is defined by $\pi(\mathcal{A},t) = \sum_{x \in L} \mu(X)(-t)^{\mathrm{rk}(X)}$.

(a) Define $\mathcal{A}$ by $Q(\mathcal{A}) = xyz(x-y)(x+y)(x-z)(x+z)(y-z)(y+z)$. Compute $\mu(X)$ for all $X \in L$ and $\pi(\mathcal{A},t)$. Use `Macaulay2` to compute the Hilbert series of $A(\mathcal{A})$.

(b) Define $\mathcal{A}$ by $Q(\mathcal{A}) = x_1 \cdots x_\ell$. Show that for $X \in L$ we have $\mu(X) = (-1)^{\mathrm{rk}(X)}$ and $\pi(\mathcal{A},t) = (1+t)^\ell$.

(3) For $X \in L(A)$ define a subarrangement $\mathcal{A}_X$ of $\mathcal{A}$ by $\mathcal{A}_X := \{H \in \mathcal{A} : X \subseteq H\}$. The *restriction* of $\mathcal{A}$ to X is the arrangement $\mathcal{A}^X$ in the vectors space X defined by

$$\mathcal{A}^X = \{X \cap H : H \in \mathcal{A} \setminus \mathcal{A}_X\}\,.$$

We call $(\mathcal{A}, \mathcal{A}', \mathcal{A}'')$ a *deletion–restriction* triple if $\mathcal{A}' = \mathcal{A} \setminus \{H_0\}$ and $\mathcal{A}'' = \mathcal{A}^{H_0}$.

(a) If $(\mathcal{A}, \mathcal{A}', \mathcal{A}'')$ is a deletion–restriction triple, then show that

$$\pi(\mathcal{A},t) = \pi(\mathcal{A}',t) + t\pi(\mathcal{A}'',t)\,.$$

Date: 27 July 2006.

(b) Let $(\mathcal{A}, \mathcal{A}', \mathcal{A}'')$ be a deletion–restriction triple. If A, A', A'' are the corresponding Orlik-Solomon algebras and $\mathrm{HS}(A,t) := \sum_p (\dim_{\mathbb{C}} A_p) t^p$, then prove that $\mathrm{HS}(A,t) = \mathrm{HS}(A',t) + t\,\mathrm{HS}(A'',t)$.

(c) Prove that $\mathrm{HS}(A,t) = \pi(A,t)$.

(d) Define $\mathcal{A}$ and $\mathcal{B}$ by $Q(\mathcal{A}) = xyz(x-z)(x+z)(y-z)(y+z)$ and

$$Q(\mathcal{B}) = xyz(x+y+z)(x+y-z)(x-y+z)(x-y-z)\,.$$

Compute $\pi(\mathcal{A},t)$ and $\pi(\mathcal{B},t)$.

(4) Consider the hyperplane arrangement $\mathcal{A}$ defined by

$$Q(\mathcal{A}) = xyzw(x+y+z)(x+y+w)(x+z+w)(y+z+w)\,.$$

(a) Construct its Eisenbud-Popescu-Yuzvinsky module $F(\mathcal{A})$ and compute its minimal free resolutions.

(b) Compute the Hilbert series of the Orlik-Solomon algebra $A(\mathcal{A})$ and compare with the betti number from part (a).

(c) Compute $\mathrm{Ext}^2_S\big(F(\mathcal{A}), S\big)$. What can you say about the support (i.e. which minimal primes lie over the annihilator)?

(d) Verify that the Orlik-Solomon ideal is generated in degree 3. What does this tell you about the linear strand of $\mathbb{Q}$ as an $A(\mathcal{A})$-module?

(5) Suppose that $Q(\mathcal{A}) = \prod_{i=1}^n \alpha_i \in R = \mathbb{C}[x_1, \dots, x_\ell]$. The *derivation module* $D(\mathcal{A})$ is the R-module of all R-derivations θ such that $\theta(Q) \in (Q)$. The Euler derivation $\sum_{i=1}^{\ell} x_i \partial/\partial x_i$ generates a free summand $R(-1)$ of $D(\mathcal{A})$ and $D(\mathcal{A}) = R(-1) \oplus D_0(\mathcal{A})$, where $D_0(\mathcal{A})$ is the kernel of the transpose of the Jacobian matrix of Q. A hyperplane arrangement $\mathcal{A}$ is *free* if and only $D(\mathcal{A})$ is a free R-module.

(a) If $Q(\mathcal{A}) = xyz(x-y)(x+y)(x-z)(x+z)(y-z)(y+z)$ then is $\mathcal{A}$ free? Compute and factor the Poincaré polynomial of $\mathcal{A}$.

(b) Show that the braid arrangement is free.

(c) Write a `Macaulay2` function that gives a matrix of generators for $D(\mathcal{A})$ as a submodule of $\mathrm{Der}(R)$. For the hyperplane arrangements in part (a) and (b), compute the determinant of this matrix.

Macaulay 2 examples from the morning lecture

```
n = 4;
E = QQ[e_1..e_n, SkewCommutative => true];
e_1*e_2 + e_2*e_1 ==  0
basis(1,E)
basis(2,E)
basis(3,E)
basis(4,E)
partial = m -> sum first entries compress diff(vars ring m, m)
```

```
partial (e_1*e_2)
partial (e_2*e_3*e_4)
partial (e_2*e_3*e_4 + e_1*e_2)

monomialSubIdeal = I -> (
     R := ring I;
     K := I;
     J := ideal(1_R);
     while (not isMonomialIdeal K) do (
          J = ideal leadTerm gens gb K;
          K = intersect(I,J));
     ideal mingens K);
orlikSolomon = method(TypicalValue => Ideal);
orlikSolomon (List, Ring) := Ideal => (A,E) -> (
     if (numgens E != #A) then error "incorrect number of variables";
     C := substitute(syz jacobian matrix{A}, E);
     M := monomialSubIdeal( ideal( (vars E) * C));
     trim ideal apply(flatten entries gens M, r -> partial r));
orlikSolomon List := Ideal => A -> (
     n := #A;
     e := symbol e;
     E := coefficientRing(ring A#0)[e_1..e_n, SkewCommutative => true];
     orlikSolomon(A, E));

R = QQ[x,y,z];
A = {x+y-z,x,y,z,x+y}
I = orlikSolomon A
reduceHilbert hilbertSeries comodule I
basis(1, comodule I)
basis(2, comodule I)
basis(3, comodule I)

x = symbol x;
l = 4;
R = QQ[x_1..x_l];
A = toList flatten apply(1..(l-1),
     i-> flatten toList apply(i+1..l, j-> {x_i-x_j}))
I = orlikSolomon A;
E = ring I;
transpose gens I
HS = reduceHilbert hilbertSeries comodule I
factor(numerator HS)
```

```
betti res (E^1/I)
J = ann(I);
transpose gens J
betti res J

S = QQ[e_1..e_#A];
sJ = substitute(ideal leadTerm J, S)
betti res sJ
sI = substitute(ideal leadTerm I, S)
betti res sI
sI == ideal dual monomialIdeal sJ

------------------------------------------------------------

partial = m -> (
     E := ring m;
     sum first entries compress diff(vars E,m));

monomialSubIdeal = I -> (
     R := ring I;
     K := I;
     J := ideal(1_R);
     while (not isMonomialIdeal K) do (
  J = ideal leadTerm gens gb K;
  K = intersect(I,J));
     ideal mingens K);

orlikSolomon = method(TypicalValue => Ideal);

orlikSolomon (List,Ring) := Ideal => (A,E) -> (
     if (numgens E != #A) then error "incorrect number of variables";
     C := substitute(syz jacobian matrix{A},E);
     M := monomialSubIdeal( ideal( (vars E) * C));
     trim ideal apply(flatten entries gens M, r -> partial r));

orlikSolomon List := Ideal => A -> (
     n := #A;
     e := symbol e;
     E := coefficientRing(ring A#0)[e_1..e_n,SkewCommutative=>true];
     orlikSolomon(A,E));
```

```
R = QQ[x,y,z] -- coordinate ring

A3 = {x,y,z,x-y,x-z,y-z}       -- A_3 braid arrangement
X3 = {x,y,z,y-z,x-z,2*x+y}
NF = {x,y,z,x-y,x-z,y-z,x+y-z} -- nonFano arrangement

HS = M -> numerator reduceHilbert hilbertSeries M;

I = orlikSolomon A3
E = ring I
A = E/I  -- Orlik Solomon algebra

apply(1..numgens(A),i->e_i=first flatten entries (vars A)_{i-1})

H = w -> ann(w)/ideal(w);

HS H(e_1)                            -- not resonant
HS H(e_1-e_6)                        -- in R^2 but not R^1
HS H(e_1-e_2)                        -- in R^1; local component
HS H(e_1+e_6+e_2+e_5-2*(e_3+e_4))  -- in nonlocal component

-- F(A)

betti res module ann orlikSolomon A3      -- linear!

symExt= (m,R) ->(
     if (not(isPolynomialRing(R))) then error "expected a polynomial ring or an exterior
     if (numgens R != numgens ring m) then error "the given ring has a wrong number of v
     ev := map(R,ring m,vars R);
     mt := transpose jacobian m;
     jn := gens kernel mt;
     q  := vars(ring m)**id_(target m);
     n  := ev(q*jn))

FA = method(TypicalValue => Module);

FA (Ideal,Ring) := Module => (j, R) -> (
     modT := (ring j)^1/(j*(ring j^1));
     F := res(prune modT, LengthLimit=>3);
     g := transpose F.dd_2;
     G := res(coker g,LengthLimit=>4);
     coker symExt(G.dd_4, R));
```

```
FA (Ideal) := Module => (j) -> (
     n := numgens ring j;
     f := symbol f;
     R := coefficientRing(ring j)[X_1..X_n];
     FA(j, R));

fa = FA orlikSolomon A3
S = ring fa
betti (FF=res fa)

lc1 = coker matrix{{X_1+X_2+X_4,X_3,X_5,X_6}}  -- S/p
apply({3,2,1,0},i-> hilbertPolynomial Tor_i(fa,lc1))
HS H (e_1-e_2) -- compare

-- this would be stupid:

FF.dd_1
-- decompose minors(6,FF.dd_1)

-- a more efficient way:

Hom(FF,S^1)
A3R1 = prune HH^2(oo)
boxList decompose ann A3R1

-- X3 example:

fa = FA orlikSolomon X3
betti (FF=res fa)
S=ring fa
X3R1 = prune Ext^2(fa,S)
boxList decompose ann X3R1 -- only local components

-- nonFano example:

fa = FA orlikSolomon NF
betti (FF=res fa)
S = ring fa
NFR1 = prune Ext^2(fa,S)
time ann NFR1;
```